

Listes

C. Charignon

Table des matières

I	Cours	2
1	Introduction	2
2	Concrètement en Caml	2
2.1	Opérations élémentaires sur tableaux et listes	2
2.2	Définition formelle	3
2.3	Filtrage	3
2.3.1	Principe	3
2.3.2	Mise en garde	3
2.3.3	Deuxième mise en garde et clauses gardées	4
2.3.4	Résumé sur les motifs	4
2.3.5	Bonus : quelques syntaxes pour gagner un peu de temps	4
2.4	Interlude méthodologique : transformer une boucle en une fonction auxiliaire récursive	5
3	Bonus : récursivité terminale	6
3.1	Intérêt de la récursivité terminale	6
II	Exercices	6

Première partie

Cours

1 Introduction

Pour enregistrer de nombreuses valeurs, la méthode la plus fréquente est le tableau (type « **Array** » en Caml). Le principe de base est simple : pour un tableau de n cases censé contenir des éléments d'un type t , on réserve en mémoire n blocs consécutifs de la taille nécessaire à enregistrer un élément de type t (64 bits pour un nombre par exemple).

Si on connaît l'adresse du début du tableau, alors pour tout $i \in \llbracket 0, n - 1 \rrbracket$, le contenu de la i -ème case est situé à l'adresse : « début du tableau + $i \times$ taille d'un bloc ».

On peut donc accéder directement à la i -ème case, la lire, la modifier, etc... Et cet accès est en $O(1)$ (il faut faire une multiplication).

En revanche, agrandir le tableau est compliqué : s'il ne reste pas de place après le dernier bloc, il va falloir chercher un autre endroit de la mémoire, plus grand, et y recopier tout le tableau. Soit un coût en $O(n)$.

N.B. Toutes les cases du tableau doivent avoir la même taille. C'est pourquoi Caml impose à tous les objets contenus dans le tableau d'être de même type.

Remarque : Python propose un type appelé **List**. Il s'agit de tableaux, mais modifié pour permettre d'y ajouter ou supprimer des éléments.

Nous allons maintenant découvrir un nouveau type de donnée permettant au contraire facilement de rajouter un élément. Il s'agit des listes chaînées (ou juste listes), appelé le type « **List** » en Caml.

Le fonctionnement est le suivant : on enregistre le premier élément de la liste et juste à côté l'adresse mémoire du second élément vers le second élément. À côté du second élément se trouve l'adresse du troisième, etc...

Pour rajouter un élément x c'est facile : on enregistre x et juste à côté l'adresse de l'ancien premier élément. Rien à modifier dans le reste de la liste, donc l'ajout est en $O(1)$.

Par contre, si on veut accéder au i ème élément, il faut partir du début, lire l'adresse du second élément, y aller, lire l'adresse du troisième élément, etc... Ceci fait $O(i)$ opérations.

En outre une grosse différence entre tableaux et listes : les listes sont persistantes alors que les tableaux sont modifiables. Cela rend les listes particulièrement adaptées à la programmation récursive, et les tableaux adaptés à la programmation impérative.

2 Concrètement en Caml

2.1 Opérations élémentaires sur tableaux et listes

Nous n'utiliserons pas de tableau en Caml avant plusieurs chapitres... Je note ici les commandes principales plus pour comparer les commandes utiles pour les tableaux et pour les listes.

- **Opérations élémentaires sur les tableaux :**

- ◇ créer un tableau vide de n cases contenant initialement l'élément x : `Array.make n x`
- ◇ modifier la i case : `monTableau.(i) <- nouvelleValeur`
- ◇ lire la i ème case : `monTableau.(i)`

- **Opérations élémentaires sur les listes :**

- ◇ La liste vide : `[]`
- ◇ La liste l à laquelle on rajoute l'élément x : `x::l`. L'opérateur `::` est souvent appelé « cons », car c'est le « constructeur » des listes. Par ailleurs, si $l' = x :: l$, on dit que x est la « tête » de l' , et que l en est la « queue ».

Toutes les opérations élémentaires ci-dessus s'exécutent en $O(1)$, sauf `Array.make` qui est en $O(n)$ pour créer un tableau de n cases.

Modifier le contenu d'une case d'un tableau est une procédure. La liste vide est une constante de type **List**. Les autres sont des fonctions.

Quelques autres opérations possibles sont décrites dans le document résumant les commandes Caml. Elles permettent de gagner du temps, mais toutes peuvent être redéfinies l'aide de la liste vide `[]` et du constructeur `::`. D'ailleurs si on

part par là, même la liste vide et le constructeur `:` pourront être redéfinis dès que nous auront vu comment créer des types, au prochain chapitre.

Remarque : L'opérateur `:` devrait être appelé le « constructeur des listes ». Mais comme le type des listes est le plus fréquemment utilisé dans les langages fonctionnels comme Caml, on dit souvent juste « constructeur » ou même « cons ».

On voit bien sur ces opérations élémentaires que les tableaux sont modifiable alors que les listes sont persistante. En effet il existe une opération élémentaire capable de modifier un tableau, c'est l'opération « `<-` ». Au contraire les opérations élémentaires sur les listes sont toutes des fonctions et permettent uniquement de construire de nouvelles listes.

Par exemple `x: l` renvoie une liste, c'est une expression de type liste.

Alors que `monTableau.(i)<- nouvelleValeur` modifie le tableau, mais ne renvoie rien. C'est une expression de type `unit`, ou encore une instruction.

Remarque : L'expression `x::l` renvoie une nouvelle liste, mais les données contenues dans `l` ne sont pas pour autant copiées : on crée juste une nouvelle case qui pointe vers l'ancienne liste. L'opération se fait donc en $O(1)$. Ainsi si j'effectue

```
1 let l1 = [1;2;3];;
2 let l2 = 0::l1;;
```

la partie `[1;2;3]` ne figure qu'une fois dans la mémoire. Cependant comme les listes ne sont pas modifiables, nous pouvons faire comme si les deux listes étaient parfaitement disjointes, autrement dit comme si `l2` était une copie de `l1`.

2.2 Définition formelle

Définition 2.1. Une liste peut être :

- la liste vide (`[]`);
- ou un élément ajouté devant une liste (`(un élément): (une liste)`).

Cette définition est récursive! Elle est particulièrement adapté à la programmation : nos programmes vont tout simplement suivre cette définition.

2.3 Filtrage

2.3.1 Principe

Le langage Caml fournit un mécanisme extrêmement utile pour manipuler les type construits comme les listes : le filtrage par motif, « pattern matching » en anglais. Voici un exemple :

```
1 let truc l=
2   match l with
3     | [] -> bla
4     | t::q -> blabla (* Les identifiants t et q tels que l=t::q sont automatiquement
                       ↪ créés et utilisables dans ce cas*)
```

Il faut se dire que le premier cas est atteint si `l=[]`. Et le second est atteint si $\exists t, q$ tq `l = t :: q`. Et dans ce cas, le `t` et le `q` qui existent sont automatiquement définis.

Voici quelques exemples simples :

- Calculer la longueur d'une liste (`List.length` en OCaml).
- Tester si un élément est dans une liste (`List.mem`).
- Sommer tous les éléments d'une liste d'entiers.
- Renvoyer le maximum d'une liste non vide.
- Tester si deux listes sont égales (existe déjà : opérateur `=`)
- Concaténer deux listes (existe déjà, c'est l'opérateur `@`). Préciser la complexité de cette fonction.

2.3.2 Mise en garde

De manière générale, répondre à une question de type \exists est difficile! Ne croyez pas que Caml peut faire des miracles... En réalité il ne peut effectuer cette opération que dans des cas balisés : lorsque le motif utilisé n'utilise que des constructeurs.

Par exemple :

```
1 let nimp x =
2   match x with
3     | a + b -> a
```

Il n'existe pas d'unique couple (a, b) tel que $x = a + b$, de sorte que a et b sont mal définis ici.

```
1 let nap x=
2   match x with
3     | a+1 -> a
4 ;;
```

Ici, il existe effectivement un unique a tel que $x = a + 1$. Mais ceci repose sur des propriétés de l'addition non triviales (existence d'un élément symétrique), que Caml ne connaît pas.

En fait, on ne peut utiliser dans un motif que des « constructeurs », c'est-à-dire les fonctions élémentaires utilisées pour définir un type.

Pour l'instant les constructeurs que nous avons rencontrés sont `[]`, `:`, `:`, la virgule `(,)` qui est le constructeur des couples (ou des p -uplets plus généralement).

Par exemple :

```
1 let symetrie m =
2   match m with
3     | (x,y) -> (y,x)
4 ;;
```

2.3.3 Deuxième mise en garde et clauses gardées

Deuxième mise en garde : ne pas utiliser un identifiant déjà existant dans un motif. Par exemple :

```
1 let rec appartient x l=
2   match l with
3     | x::q -> true
4     | t::q -> appartient x q
```

ne fonctionne pas. Lors de l'utilisation du motif de filtrage `x::q`, un *nouvel* identifiant x sera créé, mais Caml n'ira pas tester si ce nouveau x est égal au x de l'argument de la fonction. L'argument x est écrasé...

En conclusion : ne pas utiliser un identifiant déjà lié dans un motif de filtrage.

Voici comment corriger l'exemple précédent : on dispose d'une commande `when` pour rajouter des conditions lors d'un filtrage :

```
1 let rec cherche x l=
2   match l with
3     | t::q when t=x -> true
4     | t::q -> cherche x q
```

La condition `t=x` s'appelle une « garde ». On dit que `t::q when t=x` est une clause gardée.

Remarque : Dans la fonction ci-dessus, rien ne permet de connaître le type des éléments de l . Par conséquent, elle est utilisable avec n'importe quel type.

2.3.4 Résumé sur les motifs

En conclusion des deux paragraphes précédent :

Un motif ne peut être constitué que de constructeurs et de variables libres.

2.3.5 Bonus : quelques syntaxes pour gagner un peu de temps

- le souligné « `_` » : Si on a pas besoin d'une valeur dans un motif, on peut utiliser le souligné pour éviter d'avoir à lui attribuer un identifiant. Exemple : Si par exemple on n'a pas besoin de la valeur de `t` on peut utiliser `_` à la place :

```
1 let longueur l =
2   match l with
3     | [] -> 0
4     | _::q -> 1 + longueur q (* seul l'identifiant q est créé*)
```

Cela évite de définir un identifiant inutile et peut permettre de détecter certaines erreurs.

- la syntaxe **function** : Pour écrire une fonction qui va immédiatement filtrer l'argument qui lui est fourni, il existe une syntaxe raccourcie. Sur l'exemple précédent :

```
1 let longueur = function
2   | [] -> 0
3   | _::q -> 1 + longueur q
```

Remarquez qu'on n'écrit pas le « `match ... with` » mais également qu'on ne donne pas de nom à l'argument. Dans certains cas, cela peut être un soulagement de ne pas avoir à trouver un n -ième identifiant pour une valeur qui n'interviendrait que deux fois dans le code.

Pour une fonction ayant plusieurs arguments, on peut utiliser **function** lorsqu'on souhaite filtrer le dernier argument. Exemple :

```
1 let rec appartient x = function
2   | [] -> false
3   | t::q -> x=t || appartient x q
4 ;;
```

2.4 Interlude méthodologique : transformer une boucle en une fonction auxiliaire récursive

N'importe quel algorithme rédigé à l'aide d'une boucle peut être rédigé en remplaçant la boucle par une fonction auxiliaire récursive. Les variables de la boucle deviendront des arguments supplémentaires pour la fonction auxiliaire.

Voici un squelette de fonction Python utilisant deux variables :

```
1 def truc(args):
2   v1 = valeur_initiale_de_v1
3   v2 = valeur_initiale_de_v2
4   while c :
5     v1 = nouvelle_valeur_de_v1(v1,v2)
6     v2 = nouvelle_valeur_de_v2(v1,v2)
7
8   return resultat(v1,v2)
```

Sous forme récursive, elle devient, rédigé ici en Caml :

```
1 let truc args =
2
3   let rec aux v1 v2 =
4     if not c then resultat v1 v2
5     else
6       aux
7         (nouvelle_valeur_de_v1 v1 v2)
8         (nouvelle_valeur_de_v2 v1 v2)
9   in
10
11   aux valeur_initiale_de_v1 valeur_initiale_de_v2
12 ;;
```

Cas particulier : lorsqu'une variable contient le résultat à renvoyer (je l'appelle souvent **res** dans ce cas), on peut en général s'en passer. Sauf lorsque l'ordre dans lequel les opérations sont effectuées est important (par exemple la fonction pour renverser une liste vue en TD).

3 Bonus : récursivité terminale

3.1 Intérêt de la récursivité terminale

Considérons les deux exemples suivant pour calculer le nombre d'occurrence d'un élément dans une liste :

```
1 let rec nb0cc1 x l =
2   match l with
3     | [] -> 0
4     | t::q when t=x -> 1+ nb0cc1 x l
5 ;;
```

```
1 let rec nb0cc2Aux accu x l =
2   match l with
3     | [] -> accu
4     | t::q when t=x -> nb0cc2Aux (accu+1) x l
5 ;;
6
7 let nb0cc2 = nb0cc2Aux 0;;
```

La deuxième version utilise un argument supplémentaire qui sert d'accumulateur. Au passage, on peut préférer rédiger ainsi pour éviter de polluer l'espace des identifiants globaux avec des fonctions auxiliaires :

```
1 let nb0cc3 x l =
2   let rec aux accu = function
3     | [] -> accu
4     | t::q when t=x -> aux (accu+1) q
5     | _::q -> aux accu q
6   in
7   aux 0 l
8 ;;
```

Remarque : Quand vous créez une fonction auxiliaire, et/ou utilisez un argument supplémentaire non mentionné dans l'énoncé, il faut expliquer son rôle. Ici la description de `nb0cc2Aux` serait : « Renvoie `accu + (le nombre de x dans l)` ».

De sorte que `nb0cc2Aux 0 x l` renvoie `0 + le nombre de x dans l` , qui est bien le résultat cherché.

Maintenant, hormis les complications pour écrire le programme, qu'est-ce que cela change pour la machine ?

Et bien, la version avec accumulateur est légèrement optimisée. En effet, la pile des appels sera plus petite, elle va même toujours rester à un seul élément.

C'est du au fait que la fonction renvoie directement `nb0cc2Aux (accu+1) x q` sans autre opération derrière.

Une telle fonction est dite récursive terminale.

Remarque : Python ne détecte pas la récursivité terminale, et n'optimise donc pas la pile des appels...¹

Deuxième partie

Exercices

1. It's not a bug, it's a feature.

Exercices : listes

Pour ce TP comme pour les suivants, créez un fichier `listes.ml` dans lequel vous mettrez toutes les fonctions sur les listes qui n'existent pas déjà dans OCaml et dont vous pensez qu'elles pourront vous servir un jour. En particulier, mettez-y les fonctions venant d'exercices signalés par !!.

Exercice 1. * Représentation mémoire

Le tableau ci-dessous symbolise (de manière très simplifiée) l'état de la mémoire à un instant donné. On a représenté 50 blocs mémoire, chacun contient soit une valeur, soit un couple (valeur |# pointeur vers un autre bloc).

2.7	#31	38	#22	12	#38	-5.0	#29	15	1.3	#08	0.4	#13	39	#09	2	#21	-5	#12
01		02		03		04		05		06		07		08		09		10
-3	8	#19	2.4	#37	-6	1	[]	6.1	1.3	2	#16	-3						
11		12		13		14		15		16		17		18		19		20
9	#33	1	#45	-11	#25	1.2	3	#27	3.6	9	#10	56	-1.	#07	-7			
21		22		23		24		25		26		27		28		29		30
2.9	#34	13	0	#03	[]	[]	1.3	0.4	#01	39	#35	2	-5					
31		32		33		34		35		36		37		38		39		40
3.1	0	#23	-0.3	0	7	#42												
41		42		43		44		45		46		47		48		49		50

- On suppose que l'identificateur `liste1` contient le pointeur `#02`.
Quelle est la liste correspondant à `liste1`? Est-ce une liste d'entiers ou de flottants?
- On exécute l'instruction Caml suivante : `let liste2 = (List.tl liste1); ;`.
Quel est le pointeur que contient l'identificateur `liste2`?
- On exécute l'instruction Caml suivante : `let liste3 = 10: :liste1 ; ;`.
En supposant que Caml utilise, s'il en a besoin, la première adresse mémoire libre, quel est l'état de la mémoire de Caml à l'issue de cette instruction ?

Exercice 2. * !! Intervalle

Écrire une fonction `intervalle` prenant en argument deux entiers a et b et renvoyant la liste $\llbracket a, b \llbracket$ (dans l'ordre croissant).

Remarque : En Python, on obtient le tableau contenant les entiers de $\llbracket a, b \llbracket$ via `list(range(a,b))`.

Exercice 3. ** ! Liste retournée

- Écrire une fonction prenant une liste `l` et renvoyant la liste contenant les mêmes éléments que `l` mais dans l'ordre inverse.
Remarque : En OCaml c'est `List.rev`.
- Calculer la complexité de votre fonction.
- Si le retournement d'une liste de longueur n n'est pas $O(n^2)$, améliorez-la et retraitez cette question et la précédente.²

Exercice 4. ** ! Fonctions (à peine) plus difficiles

Donner la complexité de vos fonctions. On comptera le nombre d'utilisation du constructeur `::`, à la lecture (dans le filtrage), comme à l'écriture (construction de la liste à renvoyer).

- Renvoyer une liste privée de ses doublons.
- Prendre en entrée une liste l et un élément x et renvoyer la liste l privé de tout ses x . Puis de son premier x . Et enfin de son dernier.
- Pouvez-vous renvoyer une liste privée de son dernier x en un temps linéaire en la longueur de la liste ?
- Déterminer l'indice d'un élément x dans une liste l . On renverra une erreur si x n'est pas dans l .

Exercice 5. ** Décomposition en facteurs premiers

2. Ces deux questions sont donc récursives.

1. Écrire une fonction récursive renvoyant le plus petit diviseur ≥ 2 d'un entier.
2. Soit $n \in \mathbb{N}$ et d son plus petit diviseur ≥ 2 . Montrer que d est premier.
3. Écrire une fonction récursive prenant un entier et renvoyant la liste de ses facteurs premiers

Exercice 6. ****! Des chiffres et des lettres, simplifié**

Le but est d'écrire une fonction `sommeFait` prenant un entier n et une liste d'entiers positifs l et indiquant si on peut choisir des éléments de l dont la somme fait n . Les répétitions sont autorisées. Par exemple `sommeFait 13 [7;5;3]` renverra `true`, tandis que `sommeFait 13 [7;8]` renverra `false`. On convient que la somme d'aucun nombre vaut 0.

1. Soit l une liste non vide. Soit t, q tels que $l = t : q$. Montrer qu'on peut obtenir n comme somme d'éléments de l si et seulement si on peut obtenir n comme somme d'éléments de q ou on peut obtenir $n - t$ comme somme d'éléments de l .
2. Programmer en Caml une fonction `sommeFait` correspondante.
3. Démontrer que le programme termine et est correct. On pourra utiliser une récurrence forte sur $n + |l|$. ($|l|$ désigne ici la longueur de l .)
4. Variante : écrire une fonction qui renvoie la liste des nombres permettant d'obtenir n si c'est possible, et `[-1]` sinon.

Exercice 7. **** Base 2**

Les fonctions vues au chapitre sur la représentation des entiers sont souvent plus facilement écrites récursivement avec des listes. Dans cet exercice, un entier positif sera représenté par son écriture en base 2 « à l'envers » : le bit des unités (poids faible) en premier. Ceci permet au passage d'accroître à souhait la taille de la liste : on ne se limitera pas à 64 bits.

1. Écrire une fonction prenant en entrée deux listes d'entiers $l1$ et $l2$ représentant deux nombres écrits en binaire et renvoyant la liste contenant l'écriture en binaire de la somme des deux nombres.
2. Écrire une fonction renvoyant le produit de deux entiers écrits en base 2.
3. Écrire les fonctions de conversion base 2 vers base 10 et réciproque.

Exercice 8. ****! Tri par insertion**

Le tri par insertion est le tri du joueur de cartes. On prend une liste vide et on y insère les éléments de la liste à trier l'un après l'autre, en prenant garde de toujours maintenir une liste triée.

Programmer cette méthode, et calculer sa complexité au pire. Calculer aussi sa complexité dans le meilleur des cas.

Exercice 9. **** Ordre lexicographique**

On définit l'ordre lexicographique sur les listes d'entiers de la manière suivante : On compare les premières composantes des deux listes : la liste dont la première composante est la plus petite est considérée comme inférieure à l'autre. En cas d'égalité des premières composantes, on compare les deuxièmes composantes, et ainsi de suite jusqu'à, s'il le faut, épuisement de l'une des deux listes. Dans ce dernier cas, la liste la plus courte est considérée comme la plus petite.

Écrire une fonction Caml nommée `inferieur` qui prend comme arguments deux listes d'entiers et qui renvoie `true` si et seulement si la première liste est inférieure ou égale à la deuxième (pour l'ordre lexicographique).

Remarque : Caml propose en fait déjà cette fonction, c'est l'opérateur `<`. Cet opérateur permet également de comparer entiers, flottants... On dit que c'est un opérateur « polymorphe » car il peut utiliser différents types de données.

Exercice 10. *****! Programmation fonctionnelle**

La maîtrise de ce genre de fonctions pourra vous faire gagner beaucoup de temps...

1. On rappelle qu'un prédicat est une fonction renvoyant un booléen.
 - (a) Écrire une fonction `existe p l` prenant un prédicat et une liste et cherchant s'il existe un élément de l qui vérifie p . (En OCaml : `List.exists`.)
 - (b) De même, écrire une fonction `pourTout p l` vérifiant si le prédicat est réalisé pour tout élément de l . (En OCaml : `List.for_all`)
 - (c) Enfin, écrire une fonction `existeUnique p l` cherchant s'il existe un unique élément de l vérifiant p .
 - (d) À l'aide de la fonction `existe` et de la fonction `mem` de Caml, écrire une fonction `intersection_non_vide` qui teste si deux listes ont au moins un élément en commun.
2. Écrire une fonction `appliquee` prenant en argument une fonction f et une liste `[a0; ... ; an]`, qui renvoie la liste `[f a0; ... ; f an]`. Quel sera le type de cette fonction ?
Dans OCaml c'est : `List.map`.

3. (*exemple:*) Comment récupérer la liste des ordonnées d'une liste de couples ?
4. Écrire une fonction `appliqueeDouble` qui prend en entrée deux listes et une lci et qui renvoie la liste obtenue en appliquant la lci aux listes composante par composante.
5. Écrire une fonction `implosion` prenant en entrée une liste l , une loi de combinaison interne $*$ et un élément neutre pour $*$ et qui calcule $\star_{x \in l} x$.
Par exemple, si la loi est $+$ on obtiendra la somme, si c'est \times on obtiendra le produit. Chercher d'autres exemples.
6. Plus généralement, si e est l'élément fourni à votre fonction en lieu d'élément neutre, si x_0, \dots, x_{n-1} sont les éléments de la liste, dans l'ordre, est-ce que votre fonction calcule $((e * x_0) * x_1) \dots * x_{n-1}$ ou $(x_0 * \dots * (x_{n-2} * (x_{n-1} * e)))$?
N.B. Dans OCaml : `List.fold_left` et `List.fold_right`, selon que l'on commence les calculs par la droite ou par la gauche de la liste.
7. Réécrire une fonction `pour_tout` en se basant sur `implosion`.
8. En utilisant les questions précédentes, ainsi que la fonction `intervalle` de l'exercice 2, écrire, en une ligne, une fonction pour :
 - (a) Calculer la somme des n premiers entiers
 - (b) Calculer la somme des n premiers carrés
 - (c) Calculer la série harmonique $\sum_{k=1}^n \frac{1}{k}$
9. Écrire une fonction `filtre` prenant en entrée un prédicat p et une liste l qui renvoie la liste des éléments de l qui vérifient p .
N.B. Dans OCaml : `List.filter`
10. En déduire une fonction prenant une liste et renvoyant la somme des carrés de ses nombres positifs.
11. Écrire en une ligne une fonction calculant l'intersection de deux listes.

Quelques indications

- 2 Prendre juste le temps de se demander s'il vaut mieux procéder par récurrence sur a ou b .
- 3 Pour une version efficace : créer une fonction auxiliaire prenant un argument supplémentaire : les éléments de liste déjà traités.
- 4
 1. Utiliser `List.mem`
 - 2.
 3. Pour enlever la dernière occurrence : avec un retournement, on obtient $O(n)$. Mais la manière la plus efficace consiste à écrire une fonction qui pour tout élément x et liste l renvoie le couple $(l$ privé de son dernier $x, x \in l)$.
- 5 1) Écrire une fonction auxiliaire prenant en entrée un argument supplémentaire : le prochain entier à essayer. Ainsi :

```

1 let rec plusPetitDivAux n i =
2   (* Renvoie le plus petit diviseur de n qui soit >= i *)
3   ...

```

- 7
 1. Utiliser une fonction auxiliaire qui renvoie une valeur supplémentaire : la retenue.
 2. Écrire au préalable une fonction pour ajouter i zéros en tête d'une liste.
- 8 Écrire une première fonction `insertion` dont le but est d'insérer un élément dans une liste triée.

Quelques solutions

1

```
2
1 let rec intervalle a b=
2   (* Renvoie l'intervalle d'entiers [a,b[ *)
3   if b > a then
4     a::(intervalle (a+1) b)
5   else []
6 ;;
```

4

```
1 let rec sansLes x l=
2   (* Renvoie l privé de tous les x qu'il pourrait y avoir *)
3   match l with
4   |[] -> []
5   |t::q -> if t=x then sansLes x q
6             else t::sansLes x q
7 ;;
8
9 (* Avec une clause gardée : *)
10 let rec sansLes x l=
11   (* Renvoie l privé de tous les x qu'il pourrait y avoir *)
12   match l with
13   |[] -> []
14   |t::q when t=x -> sansLes x q
15   |t::q          -> t::sansLes x q
16 ;;
17
18 let rec sansLePremier x l=
19   (* Renvoie l privé de son premier x *)
20   match l with
21   |[] -> []
22   |t::q when t=x -> q
23   |t::q          -> t::sansLePremier x q
24 ;;
25 sansLePremier 2 [5;4;1;2;1;4;2;1;4;2];;
26
27 sansLes 2 [5;4;1;2;1;0;2];;
28
29 (* Première méthode : quand on rencontre un x, on utilise List.mem pour savoir si c'est le
   ↪ dernier *)
30 let rec sansLeDernier x = function
31 |[] -> []
32 |t::q when t=x && not (List.mem x q)-> (* Ici, t est le dernier x de la liste *)
33     q
34 |t::q -> t:: sansLeDernier x q
35 ;;
36
37
38 (* Deuxième méthode : retourner la liste *)
39 (* List.rev *)
40 let sansLeDernier x l=
41   List.rev (sansLePremier x (List.rev l));;
42 sansLeDernier 2 [5;4;1;2;5;4;1;2;1;2];;
43
44
45 (* Troisième méthode : écrire une fonction qui renvoie un couple
   (l privé de son dernier x, x appartient à l) *)
46 (* Cela revient à calculer sansLeDernier et List.mem simultanément *)
47
48
49 let rec sansLeDernierAux x l=
50   match l with
```

```

51 |[] -> [], false
52 |t::q when t=x -> let (q_sans_son_dernier_x, x_dans_q) = sansLeDernierAux x q in
53 |   if x_dans_q then t::q_sans_son_dernier_x, true
54 |   else q_sans_son_dernier_x, true
55 |t::q -> let (q_sans_son_dernier_x, x_dans_q) = sansLeDernierAux x q in
56 |   t::q_sans_son_dernier_x, x_dans_q
57 ;;
58 let sansLeDernier3 x l=
59   fst (sansLeDernierAux x l);;
60 sansLeDernier3 2 [5;0;5;6;2;5;2;1;2;3];;

```

```

5
1 let plusPetitDiviseur n =
2   let rec diviseurAux n d =
3     if n mod d = 0 then d
4     else diviseurAux n (d+1);
5   in
6     diviseurAux n 2
7 ;;
8
9 let rec facteursPremiers n =
10  match n with
11  |0 -> failwith "0 a une infinité de facteurs premiers"
12  |1 -> []
13  |(-1) -> []
14  |_ -> let d = plusPetitDiviseur n in
15  |   d :: facteursPremiers (n/d)
16 ;;
17
18 let rec enBase2 n b =
19  match n with
20  |0 -> []
21  |_ -> (n mod b) :: enBase2 (n/b) b
22 ;;

```

6. 1. Soit l une liste et $n \in \mathbb{N}$. Supposons qu'il existe (t, q) tels que $l=t : :q$. Pour obtenir n comme somme d'éléments de l , il y a uniquement les deux possibilités suivantes :
- Prendre t , puis obtenir $n - t$ comme somme d'éléments de l (on peut reprendre t car les répétitions sont autorisées ;
 - Ne pas prendre t , auquel cas il faut obtenir n comme somme d'éléments de q .

```

2.
1 let rec sommeFait n l=
2   (* Indique si on peut faire n en sommant des éléments de l (répétitions autorisées) *)
3   if n<0 then false
4   else if n=0 then true
5   else
6     match l with
7     |[] -> n=0
8     |t::q -> sommeFait (n-t) l || sommeFait n q
9   ;;

```

3. \S Pour rédiger la démonstration sans devoir utiliser des kilomètres de périphrases, il est utiles de poser quelques notations.

Notons E l'ensemble des listes d'entiers positifs. Notons pour tout $l \in E$ et tout entier n , $SF(n, l)$ le booléen « On peut obtenir n comme somme d'éléments de l ». Notre but est donc de prouver que pour tout $l \in E$ et tout entier n , $\text{sommeFait } n \text{ } l = SF(n, l)$.

Pour tout $k \in \mathbb{N}$, posons $P(k)$: « Pour toute liste l et tout entier n tel que $n + |l| = k$, on a $\text{sommeFait } n \text{ } l = SF(n, l)$ ».

- **Initialisation** : Soit $l \in E$ et $n \in \mathbb{N}$ tels que $|l| + n = 0$. Si $n < 0$, on ne peut pas obtenir n comme somme d'entiers positifs donc $SF(n, l) = \perp$ (faux). C'est aussi le résultat renvoyé par $\text{sommeFait } n \text{ } l$.
Si $n \geq 0$, alors comme $|l| \geq 0$, c'est que $n = 0$ et $l=[]$. Alors $SF(n, l) = \top$ (Vrai) car 0 est somme d'aucun nombre. C'est bien le résultat renvoyé par $\text{sommeFait } n \text{ } l$.
Ainsi, dans tous les cas, $SF(n, l) = \text{sommeFait } n \text{ } l$.

- **Hérédité** : Soit $k \in \mathbb{N}$. Supposons que $\forall l \in \llbracket 0, k \rrbracket, P(l)$. Soit $l \in E$ et $n \in \mathbb{Z}$ tel que $|l| + n = k + 1$.
 - ◇ Si $l = []$, alors $n = k + 1 > 0$ et il n'est pas possible d'obtenir n comme somme d'éléments de l . Or, `sommeFait` $\hookrightarrow n$ \perp renvoie bien \perp .
 - ◇ S'il existe t, q tel que $l = t :: q$, alors vu la première question, on a $SF(n, l) = SF(n - t, l) \vee SF(n, q)$ (\vee est le symbole mathématique pour « ou »). Or précisément, `sommeFait` n l renvoie `sommeFait` $(n-t)l \vee$ `sommeFait` n q . De plus, par hypothèse de récurrence, `sommeFait` $(n-t)l$ renvoie $SF(n-t, l)$ et `sommeFait` n q renvoie $SF(n, q)$. On obtient bien que `sommeFait` n $l = SF(n, l)$.

Ainsi par récurrence, pour tout $n \in \mathbb{N}$ $P(n)$.

```

4. let rec sommeQuiFait n l =
1  (* Entrée : n, entier
2     l, liste d'entiers positifs *)
3  (* Renvoie une liste d'éléments de l dont la somme vaut n s'il en existe, et [-1]
4     ↪ sinon. *)
5  if n < 0 then [-1]
6  else if n = 0 then []
7  else match l with
8     | [] -> [-1]
9     | t::q -> match sommeQuiFait (n-t) l with (* Essai en prenant t *)
10        | [-1] -> sommeQuiFait n q (* Dernier essai : sans prendre t *)
11        | res -> t::res
12 ;;
13 sommeQuiFait 12 [8;5;2];;
```

7

8

```

1  let rec insertion x l =
2  (* Précondition : l est triée *)
3  (* Renvoie la liste obtenue en insérant x à sa place dans l *)
4  match l with
5     | [] -> [x]
6     | t::q when x < t -> x::l
7     | t::q -> t::insertion x q
8  ;;
9  (* Cxté en  $O(|l|)$  *)
10
11 insertion 2 [1;2;8];;
12
13 let rec triInsertion l =
14 match l with
15 | [] -> []
16 | t::q -> insertion t (triInsertion q)
17 ;;
18 (* Cxté en  $O(|l|^2)$  *)
19 triInsertion [5;6;8;1;0];;
```

- Pour tout x et toute liste l , la complexité de `insertion` x l est en $O(|l|)$.
- Pour tout $n \in \mathbb{N}$, notons C_n le nombre maximal de comparaisons entre éléments de liste lors du tri d'une liste de longueur au plus n . On constate que
$$\begin{cases} C_0 = 0 \\ \forall n \in \mathbb{N}, C_n \leq (n-1) + C_{n-1} \end{cases}$$
 On en déduit classiquement que $C_n = O_{n \rightarrow \infty} (n^2)$.

9

10

```

1 let rec existe p = function
2 | [] -> false
3 | t::q when p t -> true
4 | t::q -> existe p q
5 ;;
6
```

```

7 existe (fun x -> x>0) [-9;-2;-3];;
8
9 let rec pour_tout p = function
10 |[] -> true
11 |t::q when p t -> pour_tout p q
12 |t::q -> false
13 ;;
14
15 pour_tout (fun x -> x>0) [9;2;3];;
16
17 let rec existe_unique p=function
18 |[] -> false
19 |t::q when p t -> not (existe p q)
20 |t::q -> existe_unique p q
21 ;;
22
23 existe_unique (fun x -> x>0) [-9;-2;-3];;
24
25 let intersection_non_vide l1 l2=
26 existe
27 (fun x-> List.mem x l1)
28 l2
29 ;;
30
31 intersection_non_vide [2;5;3] [9;5;0];;
32
33 let rec appliquee f =function
34 |[] -> []
35 |t::q -> f t :: appliquee f q
36 ;;
37
38 List.map snd [(1,2); (3,4)];;
39
40 let rec appliquee_double lci l1 l2 =
41 match l1, l2 with
42 |[],[] -> []
43 |[],_ -> failwith "pas la même longueur"
44 |_, [] -> failwith "pas la même longueur"
45 |t1::q1, t2::q2 -> (lci t1 t2) :: appliquee_double lci q1 q2
46 ;;
47
48 appliquee_double (+) [1;2;3] [4;5;6];;
49
50 List.fold_left;;
51 List.fold_right;;
52
53
54
55 let rec implosion lci l e =
56 match l with
57 |[] -> e
58 |t::q -> lci t (implosion lci q e)
59 ;;
60
61 (* NB : ici les calculent commencent « à droite » de la liste :
62 implosion * [x0; ... ; xk] e renvoie x0 * ... (x(k-1) * ( xk*e))
63 C'est la fonction List.fold_right de Caml *)
64
65 (* Pour commencer les calculs par la gauche, on peut utiliser un accu en argument
66 ↪ supplémentaire :
67 ( Le résultat sera le même pour les deux fonction si la lci est associative ) *)
68
69 let implosion_gauche lci e l = (* rema : le neutre sera utilisé « à gauche » de la liste *)

```

```

70 let rec aux accu = fonction
71   |[] -> accu
72   |t::q -> aux (lci accu t) q
73 in
74
75 aux e l
76 ;;
77
78
79 (*Exemple amusant : redéfinissons pour_tout *)
80 let pour_tout p l=
81   implosion
82     (&&)
83     (List.map p l)
84     true
85 ;;
86
87 (* NB : pour utiliser des fonctions ayant de gros arguments comme ici implosion, je mets
88    ↪ souvent un argument par ligne : c'est plus lisible. *)
89 pour_tout (fun x-> x>0) [4;-5;-6];;
90
91 (* Sauriez-vous faire de même pour il_existe ? *)
92
93
94 let rec intervalle a b=
95   (* Renvoie l'intervalle semi-ouvert [|a,b[ *)
96   if a>=b then []
97   else a::intervalle (a+1) b
98 ;;
99
100 let somme_premiers_entiers n=
101   implosion
102     (+)
103     (intervalle 0 n)
104     0
105 ;;
106 somme_premiers_entiers 5;;
107
108 let somme_premiers_carres n=
109   implosion
110     (+)
111     (List.map (fun x-> x*x) (intervalle 0 n) )
112     0
113 ;;
114 somme_premiers_carres 4;;
115
116 let serie_harmo n=
117   (* Attention : on manipule ici des flottants *)
118   implosion
119     (+.)
120     (List.map
121       (fun x-> 1./ float_of_int n)
122       (intervalle 0 n)
123     )
124     0.
125 ;;
126
127
128 let rec filtre p = fonction
129   |[] -> []
130   |t::q when p t -> t::filtre p q
131   |_::q -> filtre p q
132 ;;

```

```
133
134 let somme_carres_des_positifs l =
135   implosion (+)
136     (List.map
137       (fun x-> x*x)
138       (filtre (fun x-> x>0) l)
139     )
140     0
141 ;;
142
143 somme_carres_des_positifs [1;-3;4];;
144
145 let intersection l1 l2=
146   filtre (fun x-> List.mem x l1) l2;;
147
148 intersection [5;6;7] [3;4;5;1];;
```
